```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

Let us consider the acoustic equations for a compressible perfect gas (see, e.g., L. Landau, L. Lifschitz, *Physique théorique, Tome 6: Mécanique des fluides*, 2[nd] edition, MIR, 1989, p. 356). The corresponding system matrix is defined by:

```
> R:=matrix(4,4,[rho[0]*d[1],rho[0]*d[2],rho[0]*d[3],d[t]/c^2,rho[0]*d[t],
> 0,0,d[1],0,rho[0]*d[t],0,d[2],0,0,rho[0]*d[t],d[3]]);
```

$$R := \begin{bmatrix} \rho_0\, d_1 & \rho_0\, d_2 & \rho_0\, d_3 & \frac{d_t}{c^2} \\ \rho_0\, d_t & 0 & 0 & d_1 \\ 0 & \rho_0\, d_t & 0 & d_2 \\ 0 & 0 & \rho_0\, d_t & d_3 \end{bmatrix}$$

Let us introduce the ring $A = \mathbb{Q}(\rho_0, c)[d_t, d_1, d_2, d_3]$ of differential operators in d_t, d_1, d_2 and d_3 with coefficients in $\mathbb{Q}(\rho_0, c)$:

```
> A:=DefineOreAlgebra(diff=[d[1],x[1]],diff=[d[2],x[2]],diff=[d[3],x[3]],
> diff=[d[t],t],polynom=[x[1],x[2],x[3],t],comm=[rho[0],c]):
```

Let us denote by $M = A^{1\times 4}/(A^{1\times 4}\, R)$ the A-module finitely presented by the matrix R. We can now compute the endomorphism ring $E = \mathrm{end}_A(M)$.

```
> Endo:=MorphismsConstCoeff(R,R,A):
```

The A-module structure of E can be generated by means of

```
> nops(Endo[1]);
```

$$10$$

generators satisfying

```
> rowdim(Endo[2]);
```

$$14$$

A-linear relations. Let us consider the following matrix $P \in A^{4\times 4}$ of $Endo[1]$

```
> P:=Endo[1,5];
```

$$P := \begin{bmatrix} 0 & d_3 & -d_2 & 0 \\ -d_3 & 0 & d_1 & 0 \\ d_2 & -d_1 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

which defines the endomorphism f of M by $f(\pi(\lambda)) = \pi(\lambda\, P)$, where $\pi : A^{1\times 4} \longrightarrow M$ denotes the projection onto M and λ is an arbitrary element of $A^{1\times 4}$. In particular, we know that there exists a matrix $Q \in A^{4\times 4}$ satisfying the relation $R\, P = Q\, R$ defined by:

```
> Q:=Factorize(Mult(R,P,A),R,A);
```

$$Q := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & d_3 & -d_2 \\ 0 & -d_3 & 0 & d_1 \\ 0 & d_2 & -d_1 & 0 \end{bmatrix}$$

Let us compute a factorization of the system matrix R. In order to do that, we first compute a matrix defining a finite presentation of the A-module $\ker f$:

```
>  T:=KerMorphism(R,R,P,Q,A)[2,1];
```

$$T := \begin{bmatrix} d_1 & \rho_0\, d_t & -d_2 & d_3 & 0 \\ 0 & \rho_0 & 0 & 0 & \frac{d_t}{c^2} \\ -1 & 0 & 0 & 0 & d_1 \\ 0 & 0 & 1 & 0 & d_2 \\ 0 & 0 & 0 & -1 & d_3 \end{bmatrix}$$

Hence, we obtain that $\ker f \cong A^{1\times 5}/(A^{1\times 5}\, T)$ and we can easily check that $\ker f \neq 0$ which shows that the matrix R admits a non-trivial factorization which can be computed as follows:

```
>  PR:=stackmatrix(P,R): ST:=SyzygyModule(PR,A);
```

$$ST := \begin{bmatrix} \rho_0\, d_t & 0 & 0 & 0 & 0 & 0 & -d_3 & d_2 \\ d_1 & d_2 & d_3 & 0 & 0 & 0 & 0 & 0 \\ 0 & -\rho_0\, d_t & 0 & 0 & 0 & -d_3 & 0 & d_1 \\ 0 & 0 & \rho_0\, d_t & 0 & 0 & -d_2 & d_1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Then, we obtain the factorization $R = L\, S$, where the matrix S is defined by

```
>  S:=submatrix(ST,1..rowdim(ST),1..rowdim(P));
```

$$S := \begin{bmatrix} \rho_0\, d_t & 0 & 0 & 0 \\ d_1 & d_2 & d_3 & 0 \\ 0 & -\rho_0\, d_t & 0 & 0 \\ 0 & 0 & \rho_0\, d_t & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

and the matrix L by:

```
>  L:=Factorize(R,S,A);
```

$$L := \begin{bmatrix} 0 & \rho_0 & 0 & 0 & \frac{d_t}{c^2} \\ 1 & 0 & 0 & 0 & d_1 \\ 0 & 0 & -1 & 0 & d_2 \\ 0 & 0 & 0 & 1 & d_3 \end{bmatrix}$$

We can check that we have $R = L\, S$:

```
>  VERIF:=simplify(evalm(Mult(L,S,A)-R));
```

$$VERIF := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Up to signs, the matrix S can be directly obtained by means of the command *CoimMorphism*:

```
>  CoimMorphism(R,R,P,Q,A)[1];
```

$$\begin{bmatrix} -\rho_0\, d_t & 0 & 0 & 0 \\ d_1 & d_2 & d_3 & 0 \\ 0 & \rho_0\, d_t & 0 & 0 \\ 0 & 0 & -\rho_0\, d_t & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$